

```
log using d:\temp\propensity.log, replace

*****
* Propensity Score matching using psmatch2
*****

use http://www.stata-press.com/data/r13/cattaneo2, clear

ttest bweight, by(mbsmoke)
regress bweight mbsmoke
regress bweight mbsmoke mmarried c.mage##c.mage fbaby medu
psmatch2 mbsmoke mmarried c.mage##c.mage fbaby medu, out(bweight)
psmatch2 mbsmoke mmarried c.mage##c.mage fbaby medu, out(bweight) ate

*****
* Create the propensity score yourself
*****

use http://www.stata-press.com/data/r13/cattaneo2, clear
probit mbsmoke mmarried c.mage##c.mage fbaby medu
predict ps_probit
psmatch2 mbsmoke mmarried c.mage##c.mage fbaby medu, out(bweight) noreplacement

list _id ps_probit _pscore in 1/15

*****
* Assess the equivalence between treatment and untreated groups
*****

*****
* Obtained the propensity score and identify the match for each treated respondent
*****

use http://www.stata-press.com/data/r13/cattaneo2, clear
psmatch2 mbsmoke mmarried c.mage##c.mage fbaby medu, out(bweight) noreplacement
rename _id id
rename _n1 n1
rename _pscore pscore
rename _treated treated

sort id
save d:\temp\full.dta, replace

*****
* generate a data set for the treated
*****
use d:\temp\full.dta, clear
keep if treated ==1
keep id treated pscore mmarried mage fbaby medu

save d:\temp\treated.dta, replace

*****
* generate a data set for the untreated
*****
use d:\temp\full.dta, clear
keep if n1 ~=.
keep n1
rename n1 id
sort id
merge 1:1 id using d:\temp\full.dta
keep if _merge ==3
keep id treated pscore mmarried mage fbaby medu

save d:\temp\untreated.dta, replace

*****
* Appended two data sets together and assess if the treated and untreated groups are equivalent
*****

use d:\temp\treated.dta, clear
append using d:\temp\untreated.dta

ttest pscore, by(treated)
```

```
ttest mage, by( treated)
ttest medu, by( treated)
tab2 mmarried treated, chi2
tab2 fbaby treated, chi2
```

```
sort treated
by treated: sum pscore mage medu
```

```
save d:\temp\matched.dta, replace
```

```
*****
* Treatment Effect with Propensity Score Option
*****
use http://www.stata-press.com/data/r13/cattaneo2, clear
teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu, probit)
teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu, probit), gen(match)
```

```
* Matching With Multiple Neighbors
teffects psmatch (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu, probit), nn(3)
```

```
*****
* Propensity Score with Stratification
*****
use d:\temp\matched.dta, clear
reg mage treated if inrange(pscore, 0.5, 1)
reg mage treated if inrange(pscore, 0, 0.4999999)
```

```
*****
* Treatment Effect with Inverse-Probability Weighting
*****
use http://www.stata-press.com/data/r13/cattaneo2, clear
teffects ipw (bweight) (mbsmoke mmarried c.mage##c.mage fbaby medu, probit)
```

```
*****
* Treatment Effect with Augmented Inverse-Probability Weighting
*****
use http://www.stata-press.com/data/r13/cattaneo2, clear
teffects aipw (bweight prenatal1 mmarried mage fbaby) (mbsmoke mmarried c.mage##c.mage fbaby medu, probit), aequations
```

```
*****
* Treatment effects with inverse-probability-weighted regression adjustment
*****
use http://www.stata-press.com/data/r13/cattaneo2, clear
teffects ipwra (bweight prenatal1 mmarried mage fbaby) (mbsmoke mmarried c.mage##c.mage fbaby medu, probit), aequations
log close
```